

Lesson 23: Approximating the sum of a series

```
[> restart;
```

Comparison and integral tests

Let's look at one that `evalf` couldn't do until recently (it couldn't do it in Maple 11, but can in Maple 12,13 and 14).

```
> S := Sum(1/(k^2+ln(k)),k=1..infinity); evalf(S);
```

$$S := \sum_{k=1}^{\infty} \frac{1}{k^2 + \ln(k)}$$

1.584620708

```
> x:= k -> 1/(k^2 + ln(k));
```

$$x := k \rightarrow \frac{1}{k^2 + \ln(k)}$$

Does the ratio test work on this one?

```
> ratio := x(k+1)/x(k);
```

$$ratio := \frac{k^2 + \ln(k)}{(k+1)^2 + \ln(k+1)}$$

```
> limit(ratio,k=infinity);
```

1

No, it's inconclusive.

What does work is a comparison test.

If $\sum_{k=1}^{\infty} b(k)$ converges and $|x(k)| \leq b(k)$ for all $k \geq N$, then $\sum_{k=1}^{\infty} x(k)$ converges.

Moreover, we have an estimate on the tails: $\left| \sum_{k=N+1}^{\infty} x(k) \right| \leq \sum_{k=N+1}^{\infty} b(k)$.

In this case we can take $b(k) = k^{-2}$, as this is a "well-known" convergent series:

```
> sum(k^(-2),k=1..infinity);
```

$$\frac{1}{6} \pi^2$$

And it's clear that $0 < \frac{1}{k^2 + \ln(k)} \leq \frac{1}{k^2}$ for all $k \geq 1$. What's the estimate on the tail?

```
> EstimatedR := sum(k^(-2),k=N+1..infinity);
```

$$EstimatedR := \Psi(1, N+1)$$

```
> fsolve(EstimatedR = 10^(-10));
```

9.99999999 10⁹

Oops. I don't really want to add 10 billion terms to get a good value for the sum of this series. Fortunately there are other ways to do it. One way is based on the integral test. The idea is to get both good upper and lower bounds for the tail of the series, and approximate our sum using these.

Suppose $x(t)$ is a positive decreasing function for $t \geq 1$.

Then $\sum_{k=1}^{\infty} x(k)$ converges if and only if $\int_1^{\infty} x(t) dt$ converges.

Moreover, we have an estimate on the tails:


$$\int_{N+1}^{\infty} x(t) dt \leq \sum_{k=N+1}^{\infty} x(k) \leq \int_N^{\infty} x(t) dt$$

This can be seen in the following pictures (where now we're approximating a sum by integrals instead of approximating an integral by sums) :


```
> with(Student[Calculus1]):
```

```
> ApproximateInt(x(t), t=5..15, method=left, partition=10, output=plot);
```

An Approximation of the Integral of
 $f(t) = 1/(t^2 + \ln(t))$
on the Interval [5, 15]
Using a Left-endpoint Riemann Sum
Area: .1463914236




Partitions: 10




This shows that if $x(t)$ is decreasing $\sum_{k=N+1}^{\infty} x(k) \geq \int_{N+1}^{\infty} x(t) dt$.

```
> ApproximateInt(x(t), t=4..14, method=right, partition=10, output=
plot);
```

An Approximation of the Integral of
 $f(t) = 1/(t^2 + \ln(t))$
 on the Interval [4, 14]
 Using a Right-endpoint Riemann Sum
 Area: .1463914236



Partitions: 10



This shows that if $x(t)$ is decreasing $\sum_{k=N+1}^{\infty} x(k) \leq \int_N^{\infty} x(t) dt$.

The idea now is not just to make the tail small, but to use good estimates on both sides. If

$\sum_{k=N+1}^{\infty} x(k)$ is between A and B, we can use $(A+B)/2$ as an approximation for it, with error at most $(B-A)/2$. In this case

$$A = \int_{N+1}^{\infty} x(t) dt \text{ and } B = \int_N^{\infty} x(t) dt \text{ so } B-A = \int_N^{N+1} x(t) dt.$$

> `int(x(t), t=N..N+1);`

$$\int_N^{N+1} \frac{1}{t^2 + \ln(t)} dt$$

Well, that's less than what we'd get without the $\ln(t)$.

> `int(1/t^2, t=N..N+1);`

$$\left(\begin{cases} \infty & \text{And}(-1 < N, N < 0) \\ 0 & \text{otherwise} \end{cases} \right) + \frac{1}{N(N+1)}$$

Oops, better assume $N > 0$.

```
> % assuming N>0;
```

$$\frac{1}{N(N+1)}$$

```
> solve(%/2 = 10^(-10),N);
```

$$-\frac{1}{2} - \frac{1}{2} \sqrt{20000000001}, -\frac{1}{2} + \frac{1}{2} \sqrt{20000000001}$$

```
> evalf(%[2]);
```

70710.17810

OK, $N = 70711$ should do.

```
> N := 70711;
```

$N := 70711$

```
> A:= evalf(Int(x(k),k=N+1..infinity));
```

$A := 0.00001414187124$

```
> B:= evalf(Int(x(k),k=N..infinity));
```

$B := 0.00001414207124$

The next command would take a rather long time

```
> # Sapprox:= add(evalf(x(k)),k=1..N)+(A+B)/2;
```

The result would be $S_{approx} := 1.584620710$. Much faster is to use hardware floats using `evalhf`. This also should improve accuracy since it's using about 15 Digits instead of 10.

```
> Sapprox := evalhf(add(x(k),k=1..N)+(A+B)/2);
```

$S_{approx} := 1.58462070840555102$

(1.1)

```
> Digits:= 15; SMaple:= evalf(S); Sapprox - SMaple;
```

$Digits := 15$

$S_{Maple} := 1.58462070840554$

$1. 10^{-14}$

(1.2)

Estimating with the Midpoint Rule

70711 is a lot of terms to calculate. Let's see if we can do better.

The bounds we used were essentially comparing $\int_n^{n+1} x(t) dt$ to the values of x at the two endpoints:

$x(n)$ and $x(n+1)$. We're using an integral to approximate a sum, but we can relate this to what we've done before in approximating integrals by sums. And using the value of x at the left or right endpoint is a very crude way to approximate the integral. For an improvement, we might try the Midpoint Rule:

approximate $\int_{n-\frac{1}{2}}^{n+\frac{1}{2}} x(t) dt$ by $x(n)$. What's the error in this approximation? (I'll change from x to f)

temporarily, because I want something valid for arbitrary functions).

> **L:= h -> Int(f(t),t=n-h .. n + h) - 2*h*f(n);**

$$L := h \rightarrow \int_{n-h}^{n+h} f(t) dt - 2 h f(n)$$

> **L(0),D(L)(0);**

$$\int_n^n f(t) dt, 0$$

> **(D@@2)(L)(h);**

$$D(f)(n+h) - D(f)(n-h)$$

If $a < f''(t) < b$ for $n-h < t < n+h$, then $2 h a < L''(h) < 2 h b$, and integrating twice we get $\frac{h^3 a}{3} < L(h) < \frac{h^3 b}{3}$. In particular, take $h = \frac{1}{2}$ to get:

$$\frac{a}{24} < \int_{n-\frac{1}{2}}^{n+\frac{1}{2}} f(t) dt - f(n) < \frac{b}{24}$$

i.e.

$$\int_{n-\frac{1}{2}}^{n+\frac{1}{2}} f(t) dt - \frac{b}{24} < f(n) < \int_{n-\frac{1}{2}}^{n+\frac{1}{2}} f(t) dt - \frac{a}{24}$$

where $a < f''(t) < b$ for $n - \frac{1}{2} < t < n + \frac{1}{2}$. If $f''(t)$ is a decreasing function of t , we can take $a = f''\left(n + \frac{1}{2}\right)$ and $b = f''\left(n - \frac{1}{2}\right)$. Is that true for our function x ?

> **normal(diff(x(t),t\$3));**

$$\frac{2(12t^6 + 37t^4 + 21t^2 + 3 - 12t^4 \ln(t) + 3 \ln(t) + 2t^2 \ln(t) + \ln(t)^2)}{(t^2 + \ln(t))^4 t^3}$$

Yes, x'' is decreasing, at least if $t > 1$ (note that $t^2 > t > \ln(t)$, so $12t^6 - 12t^4 \ln(t) > 0$). So we have

$$\int_{n-\frac{1}{2}}^{n+\frac{1}{2}} f(t) dt - \frac{f''\left(n - \frac{1}{2}\right)}{24} < f(n) < \int_{n-\frac{1}{2}}^{n+\frac{1}{2}} f(t) dt - \frac{f''\left(n + \frac{1}{2}\right)}{24}$$

Adding this up for n from $N+1$ to infinity:

$$\int_{N+\frac{1}{2}}^{\infty} f(t) dt - \frac{\left(\sum_{n=N+1}^{\infty} f''\left(n - \frac{1}{2}\right)\right)}{24} < \sum_{n=N+1}^{\infty} f(n) <$$

$$\int_{N + \frac{1}{2}}^{\infty} f(t) dt - \frac{\left(\sum_{n=N+1}^{\infty} f''\left(n + \frac{1}{2}\right) \right)}{24}$$

To avoid having to sum the second derivatives, use integral estimates: since f'' is decreasing and

$$f'(n) \rightarrow 0, \quad \sum_{n=N+1}^{\infty} f''\left(n - \frac{1}{2}\right) < \int_{N - \frac{1}{2}}^{\infty} f''(t) dt = -f'\left(N - \frac{1}{2}\right) \text{ while}$$

$$\sum_{n=N+1}^{\infty} f''\left(n + \frac{1}{2}\right) > \int_{N + \frac{3}{2}}^{\infty} f''(t) dt = -f'\left(N + \frac{3}{2}\right). \text{ Thus}$$

$$\int_{N + \frac{1}{2}}^{\infty} f(t) dt + \frac{f'\left(N - \frac{1}{2}\right)}{24} < \sum_{n=N+1}^{\infty} f(n) < \int_{N + \frac{1}{2}}^{\infty} f(t) dt + \frac{f'\left(N + \frac{3}{2}\right)}{24}$$

This estimate is true whenever f'' is decreasing on $\left[N - \frac{1}{2}, \infty\right)$.

We want to take N large enough so the difference between the two bounds is less than $2 \cdot 10^{-10}$.

```
> fsolve((D(x)(t+3/2)-D(x)(t-1/2))/24=2*10^(-10),t=1..infinity)
;
223.092038745833
```

So $N = 224$ should do.

```
> N := 224;
N := 224
```

Here's the integral:

```
> J:= evalf(Int(x(t),t=N+1/2 .. infinity));
J:= 0.00445417368388001
```

Here are the two bounds.

```
> lowerbd:= evalf(add(x(n),n=1..N) + J + D(x)(N-1/2)/24);
lowerbd := 1.58462070830628
> upperbd:= evalf(add(x(n),n=1..N) + J + D(x)(N+3/2)/24);
upperbd := 1.58462070850307
```

Their average is our estimate.

```
> S2:= (lowerbd+upperbd)/2;
S2 := 1.58462070840468
```

This agrees quite well with the previous answer we got:

```
> Sapprox;
1.58462070840555102
```

```
> S2-SMaple;
-8.6 10^-13
```

(2.1)

Power series

Probably the most important series are power series. These have very important applications in many areas of mathematics.

A **power series** about the point $x = c$ is a series of the form

$$\sum_{n=0}^{\infty} a_n (x - c)^n = a_0 + a_1 (x - c) + a_2 (x - c)^2 + \dots$$

A power series has a **radius of convergence** R (possibly 0 or ∞): the series converges when $|x - c| < R$ and diverges when $|x - c| > R$.

The **Taylor series** of $f(x)$ about the point $x = c$ is the power series
$$\sum_{n=0}^{\infty} \frac{D^{(n)}(f)(c) (x - c)^n}{n!}$$

In the case $c = 0$ this is also called the **Maclaurin series** of $f(x)$.

The function $f(x)$ is said to be **analytic** at $x = c$ if its Taylor series about $x = c$ converges to $f(x)$ for x in some open interval around $x = c$.

Most of the functions we're familiar with are analytic, except perhaps at some exceptional points.

The partial sums of the Taylor (or Maclaurin) series of $f(x)$ are called the **Taylor (or Maclaurin) polynomials**.

Finding Taylor series

Maple has the **taylor** command to find a given number of terms of the Taylor series of an expression. You could also use **series**, which is somewhat more general (we saw it in Lesson 21).

```
> S:=taylor(exp(x), x=2, 5);
```

$$S := e^2 + e^2 (x - 2) + \frac{1}{2} e^2 (x - 2)^2 + \frac{1}{6} e^2 (x - 2)^3 + \frac{1}{24} e^2 (x - 2)^4 + O((x - 2)^5)$$

This is part of the Taylor series of e^x about $x = 2$. The third argument (5 in this case) controls how many terms you get in the result. Usually,

taylor(f(x), x=c, n) will give you the terms up to $a_{n-1} (x - c)^{n-1}$, and the rest are indicated by $O((x - c)^n)$, but not always. Here you get too few terms:

```
> A:= taylor((exp(x)-1-x)/x^2, x=0, 5);
```

$$A := \frac{1}{2} + \frac{1}{6} x + \frac{1}{24} x^2 + O(x^3)$$

Here you get a bit more than you asked for:

```
> B:= taylor(x^3*sin(x^3), x=0, 8);
```

$$B := x^6 + O(x^{11})$$

What actually happens is that **n** represents the maximum number of terms Maple uses in computing the Taylor series, and this is not necessarily the number of terms in the final result. For example, in the case of **(exp(x)-1-x)/x^2**, here's what happens. Start with

```
> taylor(exp(x), x=0, 5);
```

$$1 + x + \frac{1}{2} x^2 + \frac{1}{6} x^3 + \frac{1}{24} x^4 + O(x^5)$$

Subtract $1 + x$.


```
> taylor(% - (1+x), x=0, 5);
```

$$\frac{1}{2}x^2 + \frac{1}{6}x^3 + \frac{1}{24}x^4 + O(x^5)$$

Then divide by x^2 .

```
> taylor(%/x^2, x=0, 5);
```

$$\frac{1}{2} + \frac{1}{6}x + \frac{1}{24}x^2 + O(x^3)$$

The bottom line is, if you get fewer terms than you need, you should start with a larger n.

```
> taylor((exp(x)-1-x)/x^2, x=0, 7);
```

$$\frac{1}{2} + \frac{1}{6}x + \frac{1}{24}x^2 + \frac{1}{120}x^3 + \frac{1}{720}x^4 + O(x^5)$$

By the way, if you want a Maclaurin series, you don't need the "0":

```
> taylor((exp(x)-1-x)/x^2,x,7);
```

$$\frac{1}{2} + \frac{1}{6}x + \frac{1}{24}x^2 + \frac{1}{120}x^3 + \frac{1}{720}x^4 + O(x^5)$$

The result of **taylor** looks like an ordinary polynomial with an extra term $O((x-c)^n)$, but it's really something a bit different: a special "series" data structure. The **whattype** function tells you what type of data structure something is.

```
> whattype(S);
```

series

```
> lprint(S);
```

```
series(exp(2)+exp(2)*(x-2)+((1/2)*exp(2))*(x-2)^2+((1/6)*exp(2))*(x-2)^3+((1/24)*exp(2))*(x-2)^4+O((x-2)^5), x = 2, 5)
```

In many cases, if you want to do something with a series, you'll need the Taylor polynomial rather than this "series" structure. You can't just subtract the $O((x-c)^n)$, you must use **convert(..., polynom)**:

```
> convert(S, polynom);
```

$$e^2 + e^2(x-2) + \frac{1}{2}e^2(x-2)^2 + \frac{1}{6}e^2(x-2)^3 + \frac{1}{24}e^2(x-2)^4$$

```
> whattype(%);
```

`'+'`

(4.1)

Maple objects introduced in this lesson

taylor
whattype
convert(..., polynom)